3. A Few First Steps

2016-04-10

```
$Version
10.0 for Mac OS X x86 (64-bit) (September 10, 2014)
```

3.1 Expressions and Strings

Expressions are symbols or numbers or other mathematical entities, which are worked on in the kernel by the rules contained therein.

```
f = x
x
```

```
h = f^2
```

x^2

Strings are pure text, which is stored as such in most, but not all cases. A string is typed by quotation marks. These no longer appear in common output; they are rendered visible by the command: **Full-Form[ ]**

```
g = "f^2"
f^2
```

```
g
f^2
```

```
FullForm[g]
"f^2"
```

3.2 Operators and Symbols

`=` definition, assignment

```
For example:
In[1]:= 3
In[2]:= f = 3
```

`:=` delayed assignment (this is explained in a later chapter)

`==` equal — used in equations and logical operations
`!=` unequal — used in logical operations

`(* ... *)` — comments may be inserted anywhere in expressions.
Longer comments should be included in Text Cells:
Format -> Style -> Text

`+` addition

`-` subtraction

`*` multiplication sign, may be omitted. Ex.: `2 x = x + x`

but note: **x2** is the name of the variable `x2`, not **2 x** !!!

`/` division — Ex.: **a/b**

`^` exponentiation — Ex.: **x^2 = x*x**

```
%  preceeding output (preceeding in time, not in position in the
   notebook).

%% output before preceeding output

%4 output of  Out[4]. Attention !  The numbering gets lost as soon as
   you close the notebook. So it is not reasonable, to use it in a
   notebook you will store and use again later !

;  suppresses printing of output
```

f = 2 x^3

$2\,x^3$

f = π^5;

f

π^5

```
I  = Sqrt[-1] = imaginary unit

Pi = π  =  Ludolf's  number, a symbol!    N[π] = 3.14159 (....)

(  )  round brackets (= parentheses) for grouping terms
                                Ex.: (a + b)(a - b)

[  ]  (square brackets) for argument(s) of functions
                                Ex.: f[x], Sin[x], h[x,y]

{  }  curly brackets (= braces) for lists.

                                Ex.: {a,b,c}, {x,0,2Pi}
```

All names (of commands, functions, operators, options,...) start with upper case letters; this applies even to composed names:

Ex.: **Sin[x], Solve[], BesselJ[n,x], FindRoot[], TrigToExp[]**

In general, it is recommended to use lower case letters for the user's own expressions and names !

3.3 Some Simple Operations

1/2 + 1/3

$\frac{5}{6}$

%

$\frac{5}{6}$

f = %

$\frac{5}{6}$

f

$\frac{5}{6}$

%11

%11

```
z = 2^1000
```

10 715 086 071 862 673 209 484 250 490 600 018 105 614 048 117 055 336 074 437 503 883 703 510 511 \
249 361 224 931 983 788 156 958 581 275 946 729 175 531 468 251 871 452 856 923 140 435 984 577 \
574 698 574 803 934 567 774 824 230 985 421 074 605 062 371 141 877 954 182 153 046 474 983 581 \
941 267 398 767 559 165 543 946 077 062 914 571 196 477 686 542 167 660 429 831 652 624 386 837 \
205 668 069 376

```
(2/5)^5
```

$\frac{32}{3125}$

```
2/5^5
```

$\frac{2}{3125}$

```
0.2^5
```

0.00032

```
N[z]
```

1.07151×10^{301}

```
SetPrecision[z, 20]
```

$1.0715086071862673209 \times 10^{301}$

```
N[z, 20]
```

$1.0715086071862673209 \times 10^{301}$

An operator may also act from behind (postfix operation) by two slashes.

```
z //N
```

1.07151×10^{301}

```
%
```

1.07151×10^{301}

```
%7 + %8
```

$\frac{34}{3125}$

```
f = x
```

x

```
f^2 + 2f
```

$2 x + x^2$

```
f = Pi
```

π

```
1.1 f
```

3.45575

```
N[Pi]
```

3.14159

```
Cos[π/6]
```

$\frac{\sqrt{3}}{2}$

Mathematica knows the symbolic value for some arguments of the trigonometric functions, but not for all. If *Mathematica* cannot evaluate an input then it shows the input as output.

```
Cos[π/8]
```

$\mathrm{Cos}\left[\frac{\pi}{8}\right]$

```
Cos[π / 12]
```

$\frac{1+\sqrt{3}}{2\sqrt{2}}$

```
N[%, 12]
```

0.965925826289

```
SetPrecision[%%, 25]
```

0.9659258262890682867497431

```
Cos[%]
```

0.5686555578844953828541561

```
Cos[π / 13]
```

$\mathrm{Cos}\left[\frac{\pi}{13}\right]$

```
N[%]
```

0.970942

```
Exp[I Pi/4]
```

$e^{\frac{i\pi}{4}}$

```
N[%]
```

$0.707107 + 0.707107\, i$

```
(1 + I)/Sqrt[2]
```

$\frac{1+i}{\sqrt{2}}$

```
N[%]
```

$0.707107 + 0.707107\, i$

```
Clear[a, b]
f = (a + b) (a - b)
```

$(a-b)\,(a+b)$

```
g = Expand[f]
```

$a^2 - b^2$

```
Factor[g]
```

$(a-b)\,(a+b)$

```
g/(a - b)
```

$\frac{a^2-b^2}{a-b}$

```
Factor[%]
```

$a+b$

```
g = (a^3 - b^3)/(a - b)
```

$\frac{a^3-b^3}{a-b}$

```
Simplify[g]
```

$a^2+a\,b+b^2$

```
Factor[g]
```

$a^2+a\,b+b^2$

```
g = (a^7 - b^7)/(a - b)
```

$\frac{a^7-b^7}{a-b}$

```
Simplify[g]
```

$\frac{a^7-b^7}{a-b}$

```
Factor[g]
```

$a^6+a^5\,b+a^4\,b^2+a^3\,b^3+a^2\,b^4+a\,b^5+b^6$

```
h = g /. b -> 3
```

$\frac{-2187+a^7}{-3+a}$

g has not been changed by the operation above.

```
g
```

$\frac{a^7-b^7}{a-b}$

```
x = 3
```

3

```
f = (x + a)^2
```

$(3+a)^2$

```
Clear[x];  f = (x + a)^2
```

$(a+x)^2$

```
Expand[f]
```

$a^2+2\,a\,x+x^2$

```
Clear[f]
(* clears assignments made to f but does not remove f from the Global list. *)
```

```
f
```

f

```
Remove[f]
(* The use of this command is explaned in Chap.6  *)
```

```
LegendreP[5, x]
```

$\frac{1}{8}\left(15\,x-70\,x^3+63\,x^5\right)$

```
LegendreP[5, Cos[θ]]
```

$\frac{1}{8}\left(15\,\text{Cos}[\theta]-70\,\text{Cos}[\theta]^3+63\,\text{Cos}[\theta]^5\right)$

3.3.1 Complex Numbers

```
z = 3 + 4 I
```

3 + 4 ⅈ

```
Re[z]
```

3

```
Im[z]
```

4

```
Abs[z]
```

5

```
Conjugate[z]
```

3 - 4 ⅈ

```
Abs[z]
```

5

```
Abs[z Exp[I π / 4]]
```

5

```
Sin[z]
```

Sin[3 + 4 ⅈ]

```
Abs[%]
```

Abs[Sin[3 + 4 ⅈ]]

```
N[%]
```

27.2903

```
z = a + b I
```

a + ⅈ b

```
Abs[z]
```

Abs[a + ⅈ b]

More explanations and applications are treated in § 4.6.1 .

3.4 Exercises:

3.4.1 Compute analytically (as far as possible) and numerically the following quantities: tg(π/4), cot(π/3); log(1), log(I), log(I π).

3.4.2 Divide $x^9 - a^9$ by $x - a$; get the quotient as a polynomial.

3.4.3 Decompose $f = x^4 + 2x^3 - 13x^2 - 14x + 24$ into factors;
return to the original polynomial.

3.4.4 Get the values ot the binomial coefficients: $\binom{12}{n}$

Hint: Compute $(x+y)^{12.}$.

3.4.5 z = 2 + i π. Compute the real and the imaginary part, the conjugate to z,
|z| and the corresponding numerical values.

3.4.6 Compute the Legendrepolynomial P_7(x), replace x by cosθ and get the numerical value of the polynomial for $\theta = \pi/8$.

3.5 Simple Plots

3.5.1 Graphs in 2 Dimensions

```
Plot[x^3 - x + 1, {x,-2,2}]
```



More explanations and applications are treated in §§ 6.1.1 - 6.1.3 .

```
ParametricPlot[ {Cos[t]^2,Sin[t]^3}, {t,0,2π}]
```



More explanations and applications are treated in § 6.1.6, options in § 6.1.11.1 .

3.5.2 Perspectivic Plots of Surfaces in Three Dimensions

```
Plot3D[Sin[x y], {x, -2, 2}, {y, -2, 2}]
```



```
ParametricPlot3D[{2 Cos[t] Cos[p], 3 Cos[t] Sin[p], 4 Sin[t]}, {t, 0, π}, {p, 0, 2 π}]
```




```
Show[%, ViewPoint ->  {1,.5, - 3}]
```



More explanations and applications are treated in §§ 6.2.3, and options in 6.2.8 .

3.5.3 Exercises

3.5.1 Plot the function $f = x^2 - 5x + 1$ with the range of x chosen such that both zeros of f are shown.

3.5.2 Plot the curve x = cos(t), y = cos(2t + a) for a = 0, $\pi/3$, $\pi/2$.

3.5.3 Plot the hyperbolic paraboloid $z = \left(\frac{x}{2}\right)^2 - \left(\frac{y}{3}\right)^2$.

3.5.4 Plot the torus r = 2 + cos(t), x = r cos(p), y = r sin(p), z = sin(t).

3.6 Differentiation

```
Clear[a, x, y]
```

```
D[ Sin[a x], x]
```

a Cos[a x]

```
D[ Sin[ax], x]
```

0

```
D[ Sin[a x], {x,3}]
```

-a^3 Cos[a x]

```
D[ Sin[a x] Cosh[b y], {x,3}, {y,5}]
```

-a^3 b^5 Cos[a x] Sinh[b y]

More explanations and applications are treated in § 10.1 .

3.6.1 Exercises

3.6.1 Derive the function $f = x^2 - 5x + 1$ w.r.t. x .
3.6.2 Derive the functions $\{e^{2 a x^2} \cos(c\, x), \tan(e^{3 x}), \sin(a\, x) \cos(b\, y)\}$ w.r.t. x .
3.6.3 Derive $\left(\frac{x}{2}\right)^2 \left(\frac{y}{3}\right)^4$ w.r.t. x and y .

3.7 Integration

3.7.1 Indefinite integration:

```
Clear[a]
```

```
Integrate[ Exp[a x], x]
```

$\frac{e^{a x}}{a}$

```
Integrate[ Exp[-a x] Cos[b y], x, y]
```

$-\frac{e^{-a x} \mathrm{Sin}[b\, y]}{a\, b}$

3.7.2 Definite integral:

3.7.2.1 Analytic definite integration

```
Integrate[ Sin[a x], {x,0,π/2}]
```

$\frac{2 \mathrm{Sin}\left[\frac{a \pi}{4}\right]^2}{a}$

More explanations and applications are treated in § 10.4.1 .

3.7.2.2 Numeric integration

```
NIntegrate[Sin[3x], {x,0,0.45}]
```

0.260331

More explanations and applications are treated in § 10.4.2 .

3.7.3 Exercises

3.7.1 Compute the following definite integral in several ways and

a) Numeric integration.
b) Analytic integration, go over to numeric values with default precision and with 20 decimal places.
c) Series expansion (s. 3.8, 10 terms) of the integrand and tern by term integration

3.8 Series Expansions

```
Series[Sin[x], {x,0,5}]
```

$$x - \frac{x^3}{6} + \frac{x^5}{120} + O[x]^6$$

The bracket {} contains three arguments:

1. the expansion variable (here x),
2. the expansion point x_0 (here 0),
3. The highest power to be included (here 5).

The remainder term O[] denotes the output as part of a series, not a polynomial.

```
Series[Cot[x], {x,0,5}]
```

$$\frac{1}{x} - \frac{x}{3} - \frac{x^3}{45} - \frac{2\,x^5}{945} + O[x]^6$$

```
Series[ Sqrt[1/(1 + a x + b x^2)], {x,0,3}] //Simplify
```

$$1 - \frac{a\,x}{2} + \left(\frac{3\,a^2}{8} - \frac{b}{2}\right) x^2 + \left(-\frac{5\,a^3}{16} + \frac{3\,a\,b}{4}\right) x^3 + O[x]^4$$

```
Series[Log[x], {x,1,5}]
```

$$(x-1) - \frac{1}{2}(x-1)^2 + \frac{1}{3}(x-1)^3 - \frac{1}{4}(x-1)^4 + \frac{1}{5}(x-1)^5 + O[x-1]^6$$

```
f = Series[Cos[x], {x,0,5}]
```

$$1 - \frac{x^2}{2} + \frac{x^4}{24} + O[x]^6$$

```
g = Series[Sin[x], {x,0,5}]
```

$$x - \frac{x^3}{6} + \frac{x^5}{120} + O[x]^6$$

```
f g
```

$$x - \frac{2\,x^3}{3} + \frac{2\,x^5}{15} + O[x]^6$$

```
f/g
```

$$\frac{1}{x} - \frac{x}{3} - \frac{x^3}{45} + O[x]^4$$

```
Normal[f/g]
```

$$\frac{1}{x} - \frac{x}{3} - \frac{x^3}{45}$$

Normal removes the remainder term. Only then one can work with the series output.

```
Integrate[%, x]
```

$$-\frac{x^2}{6} - \frac{x^4}{180} + \text{Log}[x]$$

More explanations and applications are treated in § 10.2 .

3.8.1 Exercises:

3.8.1 Expand sin(sin x) around x = 0 up to the 7-th term.
3.8.2 Expand tg(x) around x = $\pi/2$ up to the third term.

3.9 Matrix Calculations

```
In[1]:= ma = {
{12,  1,  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,  0,  0},
{12, 10, -2,  2,  0,  0,  0,  0,  0,  0,  0,  0,  0,  2},
{ 0, -4,  8, -1,  0,  0,  0, -3,  1,  1,  0,  0,  0,  0},
{ 0,  2,  0, 12,  1,  0,  0,  0, -1, -1,  0,  0,  0,  0},
{12,  0,  0,  2,  8, -1, -1,  0, -1, -1,  0,  0,  0,  0},
{ 0, -2,  1,  0, -2,  6,  1,  0,  0,  2, -1,  0,  0,  0},
{ 0, -2,  1,  0, -2,  1,  6,  0,  2,  0, -1,  0,  0,  0},
{ 0,  0, -2,  0,  1,  0,  0,  6,  0,  0,  0,  0,  0,  0},
{ 0, -1,  1, -2, -1,  0,  1,  0,  8,  0,  0,  0,  2, -2},
{ 0, -1,  1, -2, -1,  1,  0,  0,  0,  8,  0,  0,  2, -2},
{ 0,  2,  0,  2,  0, -1, -1,  0,  0,  0,  6, -2, -4,  0},
{ 0,  0,  0,  0, -2,  0,  0,  0,  0,  0, -2,  4,  0, -2},
{-6,  0,  0,  0,  0,  0,  0,  0,  1,  1, -1,  0,  8,  0},
{ 0,  2,  0,  0,  0,  0,  0,  0, -1, -1,  0,  0,  0,  8}};
```

The notebook lamat.nb containing this matrix can be found on the Mathematica subwebsite in *Mathematica* course notebooks.

```
In[2]:= Det[ma] //Timing

Out[2]= {0.002366, 515 801 051 136}

In[4]:= im = Inverse[ma]  //Timing ;

In[5]:= im[[1]]

Out[5]= 0.004466

In[6]:= im[[2, 1]]
```

Out[6]= $\left\{\frac{6287}{66444}, -\frac{197}{16611}, -\frac{247}{66444}, \frac{283}{132888}, \frac{23}{132888}, -\frac{9}{44296}, -\frac{9}{44296}, -\frac{247}{132888}, \frac{15}{11074}, \frac{15}{11074}, -\frac{4}{16611}, -\frac{2}{16611}, -\frac{53}{66444}, \frac{20}{5537}\right\}$

```
Eigenvalues[N[ma]] //Timing

{0.001294, {16.7748, 12.6804, 10.8037, 10.4073, 9.22786, 8.56155,
  8.56155, 7.57739, 6.29493, 4.88859, 4.43845, 4.43845, 3.12134, 2.22371}}
```

3.9.1 Characteristic Polynomial

```
(*  B = MA  - x I,  f = Det( MA  - x I )        *)

b = ma - x IdentityMatrix[14];
f = Det[b] //Timing
```

$\{0.006098, 515801051136 - 1230268913664\, x + 1323417995264\, x^2 - 852522470400\, x^3 + 367962915328\, x^4 - 112712736896\, x^5 + 25296353536\, x^6 - 4229778464\, x^7 + 529948192\, x^8 - 49545816\, x^9 + 3404624\, x^{10} - 166832\, x^{11} + 5513\, x^{12} - 110\, x^{13} + x^{14}\}$

```
Factor[f[[2]]]
```

$(38 - 13\, x + x^2)^2\, (357202944 - 607584768\, x + 440172800\, x^2 - 179699584\, x^3 + 45999296\, x^4 - 7747200\, x^5 + 872528\, x^6 - 65080\, x^7 + 3084\, x^8 - 84\, x^9 + x^{10})$

More explanations and applications are treated in Chap.8 .

3.9.2 Exercise:

3.9.1 Compute the determinant, the inverse, the characteristic values and the

{{1, 0, -4}, {0, 5, 4}, {-4, 4, 3}}.
Factorize the characteristic polynomial.

3.10 Solving Linear Equations

3.10.1 Regular systems

```
sys = {5 x + 1 y == 3, 4 x - 3 y == 2}
```

{5 x + y == 3, 4 x - 3 y == 2}

```
Solve[ sys, {x,y}]
```

$\{\{x \to \frac{11}{19}, y \to \frac{2}{19}\}\}$

```
sol = % // Flatten
```

$\{x \to \frac{11}{19}, y \to \frac{2}{19}\}$

```
sys /. sol
```

{True, True}

```
Solve[ {5 x + 1 y == 3., 4 x - 3 y == 2}, {x,y}]
```

{{x → 0.578947, y → 0.105263}}

```
ma = {{5,1}, {4, -3}} ; MatrixForm[ma]
b  = {3,2}
mx = {x,y}
ma . mx  == b
```

$\begin{pmatrix} 5 & 1 \\ 4 & -3 \end{pmatrix}$

{3, 2}

{x, y}

{5 x + y, 4 x - 3 y} == {3, 2}

```
ma . mx  == b //Thread
```

{5 x + y == 3, 4 x - 3 y == 2}

```
Inverse[ma] . b
```

$\{\frac{11}{19}, \frac{2}{19}\}$

3.10.2 Singular systems

```
Solve[ {x - y == 1, 3 x - 3 y == 3}, {x,y} ]
```

Solve::svars: Equations may not give solutions for all "solve" variables >>

{{y → -1 + x}}

Solve::"svars" : "Equations may not give solutions for all "solve" variables " >>

```
Solve[ {x - y == 1, 3 x - 3 y == 2}, {x,y} ]
```

{}

More explanations and applications are treated in Chap.9 .

3.10.3 Exercises

3.10.1 Solve the following system of linear equations:

$x + y + z = 1$, $x + 2y + 3z = 4$, $x + 3y + 6z = 10$.

3.10.2 Solve the following system of linear equations in 2 ways:
$a x + b y = e$, $c x + d y = f$.

3.11 Solving Non-Linear Algebraic Equations

```
Clear[p2, x, a, b, c]
p2 = a x^2 + b x + c
```

$c + b x + a x^2$

```
sol =  Solve[ p2 == 0 , x]
```

$$\left\{\left\{x \to \frac{-b - \sqrt{b^2 - 4 a c}}{2 a}\right\}, \left\{x \to \frac{-b + \sqrt{b^2 - 4 a c}}{2 a}\right\}\right\}$$

```
p2 /. sol
```

$$\left\{c + \frac{b\left(-b - \sqrt{b^2 - 4 a c}\right)}{2 a} + \frac{\left(-b - \sqrt{b^2 - 4 a c}\right)^2}{4 a}, c + \frac{b\left(-b + \sqrt{b^2 - 4 a c}\right)}{2 a} + \frac{\left(-b + \sqrt{b^2 - 4 a c}\right)^2}{4 a}\right\}$$

```
% //Simplify
```

{0, 0}

```
Solve[ x^4 - 1 == 0]
```

$\{\{x \to -1\}, \{x \to -i\}, \{x \to i\}, \{x \to 1\}\}$

```
Solve[ x^4 + 1 == 0]
```

$\{\{x \to -(-1)^{1/4}\}, \{x \to (-1)^{1/4}\}, \{x \to -(-1)^{3/4}\}, \{x \to (-1)^{3/4}\}\}$

```
Solve[ x^4 + 1 == 0. ]
```

$\{\{x \to -0.707107 - 0.707107\, i\}, \{x \to -0.707107 + 0.707107\, i\},$
$\{x \to 0.707107 - 0.707107\, i\}, \{x \to 0.707107 + 0.707107\, i\}\}$

```
p3 = x^3 + x + 1
```

$1 + x + x^3$

```
sn = NSolve[p3 == 0]
```

$\{\{x \to -0.682328\}, \{x \to 0.341164 - 1.16154\, i\}, \{x \to 0.341164 + 1.16154\, i\}\}$

```
p3 /. sn
```

$\{-1.66533 \times 10^{-16}, 0. + 0.\, i, 0. + 0.\, i\}$

```
Chop[%]
```

{0, 0, 0}

s3 = Solve[p3 == 0]

$$\left\{\left\{x\to-\left(\frac{2}{3\left(-9+\sqrt{93}\right)}\right)^{1/3}+\frac{\left(\frac{1}{2}\left(-9+\sqrt{93}\right)\right)^{1/3}}{3^{2/3}}\right\},\left\{x\to-\frac{\left(1+i\sqrt{3}\right)\left(\frac{1}{2}\left(-9+\sqrt{93}\right)\right)^{1/3}}{2\times3^{2/3}}+\frac{1-i\sqrt{3}}{2^{2/3}\left(3\left(-9+\sqrt{93}\right)\right)^{1/3}}\right\},\left\{x\to-\frac{\left(1-i\sqrt{3}\right)\left(\frac{1}{2}\left(-9+\sqrt{93}\right)\right)^{1/3}}{2\times3^{2/3}}+\frac{1+i\sqrt{3}}{2^{2/3}\left(3\left(-9+\sqrt{93}\right)\right)^{1/3}}\right\}\right\}$$

Simplify[s3]

$$\left\{\left\{x\to\frac{-2\left(\frac{3}{-9+\sqrt{93}}\right)^{1/3}+\left(2\left(-9+\sqrt{93}\right)\right)^{1/3}}{6^{2/3}}\right\},\left\{x\to-\frac{\left(1+i\sqrt{3}\right)\left(\frac{1}{2}\left(-9+\sqrt{93}\right)\right)^{1/3}}{2\times3^{2/3}}+\frac{1-i\sqrt{3}}{2^{2/3}\left(3\left(-9+\sqrt{93}\right)\right)^{1/3}}\right\},\left\{x\to\frac{i\left(i+\sqrt{3}\right)\left(\frac{1}{2}\left(-9+\sqrt{93}\right)\right)^{1/3}}{2\times3^{2/3}}+\frac{1+i\sqrt{3}}{2^{2/3}\left(3\left(-9+\sqrt{93}\right)\right)^{1/3}}\right\}\right\}$$

N[%]

$\{\{x\to-0.682328\},\{x\to0.341164-1.16154\,i\},\{x\to0.341164+1.16154\,i\}\}$

p3 /. s3

$$\left\{1-\left(\frac{2}{3\left(-9+\sqrt{93}\right)}\right)^{1/3}+\frac{\left(\frac{1}{2}\left(-9+\sqrt{93}\right)\right)^{1/3}}{3^{2/3}}+\left(-\left(\frac{2}{3\left(-9+\sqrt{93}\right)}\right)^{1/3}+\frac{\left(\frac{1}{2}\left(-9+\sqrt{93}\right)\right)^{1/3}}{3^{2/3}}\right)^3,\right.$$
$$1-\frac{\left(1+i\sqrt{3}\right)\left(\frac{1}{2}\left(-9+\sqrt{93}\right)\right)^{1/3}}{2\times3^{2/3}}+\frac{1-i\sqrt{3}}{2^{2/3}\left(3\left(-9+\sqrt{93}\right)\right)^{1/3}}+\left(-\frac{\left(1+i\sqrt{3}\right)\left(\frac{1}{2}\left(-9+\sqrt{93}\right)\right)^{1/3}}{2\times3^{2/3}}+\frac{1-i\sqrt{3}}{2^{2/3}\left(3\left(-9+\sqrt{93}\right)\right)^{1/3}}\right)^3,$$
$$\left.1-\frac{\left(1-i\sqrt{3}\right)\left(\frac{1}{2}\left(-9+\sqrt{93}\right)\right)^{1/3}}{2\times3^{2/3}}+\frac{1+i\sqrt{3}}{2^{2/3}\left(3\left(-9+\sqrt{93}\right)\right)^{1/3}}+\left(-\frac{\left(1-i\sqrt{3}\right)\left(\frac{1}{2}\left(-9+\sqrt{93}\right)\right)^{1/3}}{2\times3^{2/3}}+\frac{1+i\sqrt{3}}{2^{2/3}\left(3\left(-9+\sqrt{93}\right)\right)^{1/3}}\right)^3\right\}$$

Together[%]

$\{0,0,0\}$

p5 = x^5 - x + 1

$1-x+x^5$

```
s5 = Solve[p5 == 0]
```

{{x → Root[1 - #1 + #1^5 &, 1]}, {x → Root[1 - #1 + #1^5 &, 2]},
{x → Root[1 - #1 + #1^5 &, 3]}, {x → Root[1 - #1 + #1^5 &, 4]}, {x → Root[1 - #1 + #1^5 &, 5]}}

The expression ending with the empercent sign is a "pure function" .
Hashmarks (#, #1, ...) represent the variables.

```
ns5 = N[%]
```

{{x → -1.1673}, {x → -0.181232 - 1.08395 ⅈ}, {x → -0.181232 + 1.08395 ⅈ},
{x → 0.764884 - 0.352472 ⅈ}, {x → 0.764884 + 0.352472 ⅈ}}

```
p5 /. ns5
```

$\{-4.44089\times10^{-16}, -4.44089\times10^{-16} - 2.22045\times10^{-16}\,i, -4.44089\times10^{-16} + 2.22045\times10^{-16}\,i,$
$2.22045\times10^{-16} + 5.55112\times10^{-17}\,i, 2.22045\times10^{-16} - 5.55112\times10^{-17}\,i\}$

More explanations and applications are treated in Chap.9 .

Exercises

3.11.1 Solve the following algebraic equation (analytically and numerically):
$x^5 + x - 1 = 0$. Verify by inserting that the solutions fulfil the equation.

3.11.2 Solve the following algebraic equation (analytically and numerically):
$x^8 - 13 x^4 + 7 = 0$. Verify by inserting that the solutions fulfil the equation.

3.12 Finding Roots of Algebraic and Transcendental Equations

```
Clear[a, x, y, z]
```

```
p3 = x^3 + x + 1
```

$1 + x + x^3$

```
s1 = FindRoot[p3, {x,1}]
```

{x → -0.682328}

```
s2 = FindRoot[p3, {x,I}]
```

{x → 0.341164 + 1.16154 ⅈ}

```
s3 = FindRoot[p3, {x,-I}]
```

{x → 0.341164 - 1.16154 ⅈ}

```
so = {s1, s2, s3}
```

{{x → -0.682328}, {x → 0.341164 + 1.16154 ⅈ}, {x → 0.341164 - 1.16154 ⅈ}}

```
p3 /. so
```

$\{-9.4369\times10^{-16}, 0. + 0.\,i, 0. + 0.\,i\}$

```
Solve[Cos[x] == 0, x]
```

$\{\{x \to \text{ConditionalExpression}[-\frac{\pi}{2} + 2\pi C[1], C[1] \in \text{Integers}]\},$
$\{x \to \text{ConditionalExpression}[\frac{\pi}{2} + 2\pi C[1], C[1] \in \text{Integers}]\}\}$

Message in Message Window:

```
Solve::ifun :
   Inverse functions are being used by Solve, so some solutions may not be found;
use Reduce for complete solution information. >>
```

```
? Reduce
```

Reduce[*expr*, *vars*] reduces the statement *expr* by solving equations or inequalities for *vars* and eliminating quantifiers.
Reduce[*expr*, *vars*, *dom*] does the reduction over the domain *dom*. Common choices of *dom* are Reals, Integers, and Complexes. >>

```
Reduce[Cos[x] == 0, x]
```

$C[1] \in \text{Integers} \&\& \left(x == -\frac{\pi}{2} + 2\pi C[1] \,||\, x == \frac{\pi}{2} + 2\pi C[1]\right)$

```
soq = Solve[Sin[x] + Cos[x] == 0, x]
```

$\left\{\left\{x \to \text{ConditionalExpression}\left[-\frac{\pi}{4} + 2\pi C[1],\ C[1] \in \text{Integers}\right]\right\},\ \left\{x \to \text{ConditionalExpression}\left[\frac{3\pi}{4} + 2\pi C[1],\ C[1] \in \text{Integers}\right]\right\}\right\}$

Message in Message Window as above.

```
FindRoot[hh[x_] = Sin[x] + Cos[x] , {x, 2.1} ]
```

{x → 2.35619}

$N\left[-\frac{\pi}{4} + 2\pi\,\text{Range}[-1, 1]\right]$

{-7.06858, -0.785398, 5.49779}

$N\left[-\frac{\pi}{4} + 2\pi\,\text{Range}[-1, 1, 1/2]\right]$

{-7.06858, -3.92699, -0.785398, 2.35619, 5.49779}

```
hh[%]
```

$\{3.33067\times10^{-16}, -2.22045\times10^{-16}, 1.11022\times10^{-16}, 1.11022\times10^{-16}, -3.33067\times10^{-16}\}$

```
Plot[hh[x], {x, -1, 2.5}]
```




```
Solve[{x^3 - y^2 == 0, x + y == 2}, {x, y}]
```

{{x → -2 i, y → 2 + 2 i}, {x → 2 i, y → 2 - 2 i}, {x → 1, y → 1}}

```
ContourPlot[{x^3 - y^2 == 0, x + y == 2}, {x, -2, 2}, {y, 0, 2}]
```




```
FindRoot[{x^3 - y^2, x + y - 2}, {x, 1}, {y, 2}]
```

{x → 1., y → 1.}

```
FindRoot[{x^3 - y^2, x + y - 2}, {x, I}, {y, 2 I}]
```

$\{x \to 3.6393 \times 10^{-17} - 2.\, i, y \to 2. + 2.\, i\}$

```
f = z^2 + Conjugate[z]
```

z^2 + Conjugate[z]

```
FindRoot[f, {z, 1}]
```

$\{z \to 3.52718 \times 10^{-18}\}$

```
FindRoot[f, {z, 1, I}]
```

{z → 0.5 + 0.866025 i}

One may give also two starting vales. In some cases FindRoot[] even asks for this.

More explanations and applications are treated in Chap.9 .

Exercises

3.12.1 Find the numeric values of two other roots of the function:
f = sin(x)-cos(x).

3.12.2 Find all the numeric values x in the interval (-3, 3) for which the function
f = (tg x - x)/($\frac{x}{3}$ + cos x) becomes either 0 oder infinite.

3.13 Solution of Differential Equations

3.13.1 Analytic Solution : General Solution

' (apostrophe) denotes the derivative w.r.t. the independent variable

```
Clear[k, x, y]
```

```
DSolve[ y''[x] + k y[x] == 0, y[x], x]
```

$\{\{y[x] \to C[1] \operatorname{Cos}[\sqrt{k}\ x] + C[2] \operatorname{Sin}[\sqrt{k}\ x]\}\}$

```
DSolve[ y''[x] + 2 y'[x]/x - l (l + 1)/x^2 y[x] +
        k^2 y[x] == 0, y[x], x]
```

{{y[x] → C[1] SphericalBesselJ[l, k x] + C[2] SphericalBesselY[l, k x]}}

3.13.2 Analytic Solution : Initial Value Problem

```
Clear[t,x,y,g];
DSolve[ {x''[t] == 0, y''[t] + g == 0, x[0] == 0,
x'[0] == v0, y[0] == 5, y'[0] == 0}, {x[t], y[t]}, t]
```

$\{\{x[t] \to t\, v0,\ y[t] \to \frac{1}{2}(10 - g\, t^2)\}\}$

More explanations and applications are treated in §§ 11.1 - 11.4 .

3.13.3 Numerical Solution

Motion of mass point in gravity field with and without Newtonian friction.

```
Clear[r, v, b, t, x, y]
```

Position

```
r[t_] = {x[t], y[t]}
```

{x[t], y[t]}

Velocity

```
v[t_] = D[r[t],t]
```

$\{x'[t],\ y'[t]\}$

Acceleration

```
b[t_] = D[v[t],t]
```

$\{x''[t],\ y''[t]\}$

m b = F = m g

```
m = 1; g = 10; a = .3;
sys = m b[t] == {0, -m g}
```

$\{x''[t],\ y''[t]\} == \{0, -10\}$

```
sys = sys //Thread
```

$\{x''[t] == 0,\ y''[t] == -10\}$

m b = F = m g - a v |v|

```
sysa = m b[t] ==
{0, -m g} - a v[t] Sqrt[x'[t]^2 + y'[t]^2] //Thread
```

$\{x''[t] == -0.3\, x'[t] \sqrt{x'[t]^2 + y'[t]^2},\ y''[t] == -10 - 0.3\, y'[t] \sqrt{x'[t]^2 + y'[t]^2}\}$

```
anf = {x[0] == 0, y[0] == 0, x'[0] == 2, y'[0] ==10}
```

$\{x[0] == 0,\ y[0] == 0,\ x'[0] == 2,\ y'[0] == 10\}$

```
sol = NDSolve[ Join[sys,anf], {x,y},
{t,0,4}] //Flatten
```

```
{x → InterpolatingFunction[ Domain {{0., 4.}} Output scalar ],
 y → InterpolatingFunction[ Domain {{0., 4.}} Output scalar ]}
```

```
sola = NDSolve[ Join[sysa,anf], {x,y},
{t,0,2}] //Flatten
```

```
{x → InterpolatingFunction[ Domain {{0., 2.}} Output scalar ],
 y → InterpolatingFunction[ Domain {{0., 2.}} Output scalar ]}
```

```
p = ParametricPlot[{x[t], y[t]} /.sol, {t, 0, 2}];
pa =
  ParametricPlot[{x[t], y[t]} /.sola, {t, 0, 1.35`}, PlotStyle → Dashing[{0.01`}]];
```

```
Show[p, pa]
```



More explanations and applications are treated in § 11.2 .

Exercise

3.13.1 Solve the 2-dimensional equation of motion for the motion of a projectile experiencing the field of gravity and a linear friction force.

3.14 Vector Algebra and Analysis in Orthogonal Curvilinear Coordinates

In *Mathematica* versions up to version 8 the vector analysis functionality is contained in an own package named "VectorAnalysis`", which must be loaded as shown in the next *Mathematica* command just below. Since version 9 onwards the main commands are contained in the kernel. These new commands are more involved than those of the earliner versions. However, the commands of the previous versions can still be used if the package "VectorAnalysis`" is loaded.
This loading overrides the new functionality and provides almost complete compatibility for programs written in the old verions. Here I shall describe the old version, so I use the package "VectorAnalysis`".
I shall deal with the new version in Chap.14 .

```
Needs["VectorAnalysis`"]
```

3.14.1 Vector algebra

Cartesian coordinates x, y, z

There are three vectors $\vec{a}$, $\vec{b}$, $\vec{c}$:

```
Clear[va, vb, vc]
```

```
 va = {ax, ay, az} ;
 vb = {bx, by, bz};
vc  = {cx, cy, cz} ;
```

Scalar product:

```
α = va.vb
```

```
ax bx + ay by + az bz
```

Vector product:

```
vcc = Cross[va, vb]
```

```
{-az by + ay bz, az bx - ax bz, -ay bx + ax by}
```

Scalar triple product (Ge.: Spatprodukt)

```
β = vc.Cross[va, vb]
```

```
(-az by + ay bz) cx + (az bx - ax bz) cy + (-ay bx + ax by) cz
```

Spherical coordinates r, θ, ϕ

There are three vectors $\vec{a}$, = $\vec{e}$r ar + $\vec{e}\theta$ aθ + $\vec{e}\phi$ aϕ ; $\vec{b}$, $\vec{c}$ defined analogously.

```
Clear[va, vb, vc]
```

```
 va = {ar, aθ, aϕ} ;
 vb = {br, bθ, bϕ} ;
vc  = {cr, cθ, cϕ} ;
```

Scalar product:

```
α = va.vb
```

```
ar br + aθ bθ + aϕ bϕ
```

Vector product:

```
vcc = Cross[va, vb]
```

```
{-aϕ bθ + aθ bϕ, aϕ br - ar bϕ, -aθ br + ar bθ}
```

Scalar triple product (Ge.: Spatprodukt)

```
β = vc.Cross[va, vb]
(-aϕ bθ + aθ bϕ) cr + (aϕ br - ar bϕ) cθ + (-aθ br + ar bθ) cϕ
```

In all orthogonal curvilinear physical coordinate systems, vectors are set up as linear combinations of the base vectors; in *Mathematica* they are written as lists comprising the three components.

3.14.2 Coordinate transformations

Transformatons between Cartesian coordinates x, y, z and spherical coordinates r, θ, ϕ

```
Needs["VectorAnalysis`"]

SetCoordinates[Spherical[r, θ, ϕ]]
Spherical[r, θ, ϕ]

CoordinateSystem
Spherical

CoordinatesToCartesian[{r, θ, ϕ}]
{r Cos[ϕ] Sin[θ], r Sin[θ] Sin[ϕ], r Cos[θ]}

CoordinatesToCartesian[{4.5, π / 6., π / 4.}]
{1.59099, 1.59099, 3.89711}

CoordinatesFromCartesian[{x, y, z}]
```

$$\left\{\sqrt{x^2+y^2+z^2},\ \mathrm{ArcCos}\left[\frac{z}{\sqrt{x^2+y^2+z^2}}\right],\ \mathrm{ArcTan}[x, y]\right\}$$

```
CoordinatesFromCartesian[
 {1.5909902576697317`, 1.5909902576697315`, 3.897114317029974`}]
{4.5, 0.523599, 0.785398}

{4.5, π / 6., π / 4.}
{4.5, 0.523599, 0.785398}
```

3.14.3 Vector analysis in orthogonal curvilinear coordinates

Cartesian coordinates

```
Needs["VectorAnalysis`"]

SetCoordinates[Cartesian[x, y, z]]
Cartesian[x, y, z]

CoordinateSystem
Cartesian

Grad[ψ[x,y,z],{x,y,z} ]
```

$$\left\{\psi^{(1,0,0)}[x, y, z],\ \psi^{(0,1,0)}[x, y, z],\ \psi^{(0,0,1)}[x, y, z]\right\}$$

```
Div[ {ax[x,y,z], ay[x,y,z], az[x,y,z]},{x,y,z} ]
```

$$az^{(0,0,1)}[x, y, z] + ay^{(0,1,0)}[x, y, z] + ax^{(1,0,0)}[x, y, z]$$

```
Curl[ {ax[x,y,z], ay[x,y,z], az[x,y,z]},{x,y,z} ]
```

$\{-ay^{(0,0,1)}[x, y, z] + az^{(0,1,0)}[x, y, z],$
$ax^{(0,0,1)}[x, y, z] - az^{(1,0,0)}[x, y, z], -ax^{(0,1,0)}[x, y, z] + ay^{(1,0,0)}[x, y, z]\}$

```
Laplacian[ψ[x, y, z], {x, y, z}]
```

$\psi^{(0,0,2)}[x, y, z] + \psi^{(0,2,0)}[x, y, z] + \psi^{(2,0,0)}[x, y, z]$

Spherical coordintes

```
SetCoordinates[Spherical[r,θ,ϕ]]
```

Spherical[r, θ, ϕ]

```
Grad[ ψ[r,θ,ϕ],{r,θ,ϕ},"Spherical" ]
```

$$\left\{\psi^{(1,0,0)}[r, \theta, \phi], \frac{\psi^{(0,1,0)}[r, \theta, \phi]}{r}, \frac{\text{Csc}[\theta]\, \psi^{(0,0,1)}[r, \theta, \phi]}{r}\right\}$$

```
Div[ {ar[r,θ,ϕ], ath[r,θ,ϕ], aph[r,θ,ϕ]},{r,θ,ϕ},"Spherical" ] //
Expand
```

$$\frac{2\, ar[r, \theta, \phi]}{r} + \frac{ath[r, \theta, \phi]\, \text{Cot}[\theta]}{r} + \frac{\text{Csc}[\theta]\, aph^{(0,0,1)}[r, \theta, \phi]}{r} + \frac{ath^{(0,1,0)}[r, \theta, \phi]}{r} + ar^{(1,0,0)}[r, \theta, \phi]$$

```
Curl[{ar[r,θ,ϕ], ath[r,θ,ϕ], aph[r,θ,ϕ]},{r,θ,ϕ},"Spherical" ] //
Expand
```

$$\left\{\frac{aph[r, \theta, \phi]\, \text{Cot}[\theta]}{r} - \frac{\text{Csc}[\theta]\, ath^{(0,0,1)}[r, \theta, \phi]}{r} + \frac{aph^{(0,1,0)}[r, \theta, \phi]}{r},\right.$$
$$-\frac{aph[r, \theta, \phi]}{r} + \frac{\text{Csc}[\theta]\, ar^{(0,0,1)}[r, \theta, \phi]}{r} - aph^{(1,0,0)}[r, \theta, \phi],$$
$$\left.\frac{ath[r, \theta, \phi]}{r} - \frac{ar^{(0,1,0)}[r, \theta, \phi]}{r} + ath^{(1,0,0)}[r, \theta, \phi]\right\}$$

```
Laplacian[ ψ[r,θ,ϕ],{r,θ,ϕ} ,"Spherical" ] //Expand
```

$$\frac{\text{Csc}[\theta]^2\, \psi^{(0,0,2)}[r, \theta, \phi]}{r^2} + \frac{\text{Cot}[\theta]\, \psi^{(0,1,0)}[r, \theta, \phi]}{r^2} + \frac{\psi^{(0,2,0)}[r, \theta, \phi]}{r^2} + \frac{2\, \psi^{(1,0,0)}[r, \theta, \phi]}{r} + \psi^{(2,0,0)}[r, \theta, \phi]$$

More explanations and applications are treated in Chap.14 .

Exercises

3.14.1 Go over to circular cylindrical coordinates r, ϕ, z
(SetCoordinates[Cylindrical[r,ϕ,z]]).
Seek the transformations between the Cartesian and the cylindrical coordinates of the same point (x,y,z) ⇔ (r, ϕ, z) . For the point x = 2.3, y = 3.7, z = 1.4 find the values of the corresponding cylindrcal coordinates. Transform them back to the Cartesian components.
Compute the gradient and Laplacian of a scalar function ψ(r,ϕ,z); divergence and curl of a vector field (ar(r,ϕ,z), ap(r,ϕ,z), az(r,ϕ,z)). Do not forget to insert the name "Cylindrical" into the operators of vector analysis.

3.14.2 Verify in rectangular, spherical and cylindrical coordinates that

```
Laplacian ψ  =  div grad ψ.
```